T.C.
YILDIZ TECHNICAL UNIVERSITY
FACULTY OF MECHANICAL ENGINEERING
DEPARTMENT OF INDUSTRIAL ENGINEERING

Industrial Eng. Design 2 Project Assignment

Orhan ŞÖFÖRTAKIMCI
17061073

ALPER EKENER
17061005

Adviser
Assoc. Prof. Vildan ÖZKIR

İSTANBUL, 2022

PREFACE

We would like to thanks Assoc. Prof. Vildan ÖZKIR, who did not spare us his valuable experience while preparing the Industrial Eng. Design 2 Project Assignment and contributed greatly to our work by providing us with data and information support so that we could carry out the application part of our study.

January, 2022

Orhan ŞÖFÖRTAKIMCI

Alper EKENER

TABLE OF CONTENTS

ABBREVIATON

CSPL: The Charging Station Placement

EV: Electric Vehicles

EVCS: Electric Vehicles Charging Station

GAMS: General Algebraic Modeling System

MIT: Mixed Integer Programming

RNC: Route Node Covering

FIGURE LIST

Location Optimization of Electric Vehicle Charging Stations

Alper EKENER

Orhan ŞÖFÖRTAKIMCI

Department of Industrial Engineering

Industrial Eng. Design 2 Project Assignment

Adviser: Assoc. Prof. Vildan ÖZKIR

Nowadays, with the rapid depletion of fossil fuels and developing technology, the interest in electric vehicles (EV) is increasing day by day. As in the world, the use of EV in our country is increasing day by day. With TOGG, which is a domestic production project, it is thought that many more people will own EV in the coming years. As the number of vehicles increased, the need for electric vehicle condition unit increased. In this study, a model for the location optimization of EV charging stations has been developed. As an example, the candidate stations to be located in Istanbul are used. With the model, it has been decided in which of the candidate locations the charging station will be established. This MIT model was solved with the GAMS program and the optimum result was achieved.

Keywords: Optimization, Electric Vehicle, Charge Station, MIT model

CHAPTER 1

1. INTRODUCTION

With the prominent problem of sustainable development and utilization of fossil energy, the environmental and development problems caused by the greenhouse effect are becoming more and more serious. Countries around the world are actively fulfilling their energy conservation and emission reduction commitments [1] .

Due to the world's shortage of fossil fuels and the serious environmental pollution from burning them, seeking alternative energy has become a crucial topic of research. Transportation is one of the main consumers of energy and contributors to air pollution. Electric Vehicles (EVs) move pollution away from urban areas and electricity can be efficiently transformed from both traditional fossil fuels and promising renewable energies like solar energy and tidal energy. EVs, as a replacement of traditional internal combustion engine vehicles, provides an environment-friendly solution to modern cities' transportation. A rapid growth of EVs has been seen in recent years along with the rising popularity of the notion of smart cities [2].

While the average fueling times of conventional vehicles vary between a few days and a week, EVs need to be recharged much more frequently due to insufficient battery size. In addition, the EV charging process is much longer than the fuel filling times of conventional vehicles. Therefore, an effective charging infrastructure planning is required to meet EV demands. When making an effective planning, the fact that EVs will use charging stations very often and long charging times should be taken into account. In addition, vehicle driving profiles and electrical network infrastructure should be considered for a determined region and optimal planning should be carried out [3].

The number of electric vehicles is increasing day by day in our country, Turkey, as in other countries. Most of the EVs are located in Istanbul. Together with our domestic project TOGG, which has been on the agenda recently, this number will be exponentially greater in the near future. Therefore, extensive project studies are required. EV owners should reach the charging stations as soon as possible, the charging station should be empty at that time. In terms of cost, the number of charging stations should also be minimal.

1.1. Literature Review

Many approaches have been proposed to address the optimization problem of allocating EVCSs. Proposed models in the literature can be categorized into flow-capturing models and node-serving models. EVCS allocation is a challenging problem and is extremely crucial to the EV industry. Although flow-capturing models are the most popular solutions proposed in literature. These models suffer from inflexibility, highcomplexity, and centrality. On the other hand, node-serving models such as the models proposed by Vazifeh et al. [4], Rahman et al. [5], and Vasant et al. [6] showed better flexibility and were able to simulate realistic scenarios for an EV network. However, none of these models captured the uncertainty of EV demand. Estimating EV demand has been studied using different methods in the literature. The demand uncertainty is captured using Markov-chain model. However, none of the proposed models to estimate demand uncertainty using Markov-chain were utilized in allocating EVCS.

Apart from these, there are some optimum route selection models in the literature. In addition, there are various studies on determining the location and capacity of stations. An iterative approach technique is presented to solve the Route Node Covering (RNC) problem [7]. The purpose of the RNC problem is to find the minimum number of charging stations and their locations. Queuing theory has also been used to arrange by charging stations in a way that minimizes drivers' queue time.

A realistic CSPL model [8] is proposed that takes into account the strategic behavior of electric vehicle drivers to minimize charging costs. In addition, the traffic situation and queue time at charging stations, which play a vital role in EV propagation, are also taken into account in our model. The OCEAN algorithm is proposed to calculate the optimal solution. Unfortunately OCEAN is not scalable due to integer variables and large search space. Therefore, an effective heuristic OCEAN-C algorithm is proposed to speed up computation and solve realistic CSPL problems.

There are also studies on route optimization for electric vehicles to use their charge effectively. [9].

According to the researches, many studies have been carried out on the location optimization of electric vehicles. It has been seen that MIT is at the forefront in evaluation methods.

1.2. The Goal of The Project

Most of the electric vehicle users are located in Istanbul. This number is expected to increase further in the coming years. With this project, it is aimed to correctly position the candidate charging stations in order to prevent the electric vehicle charging stations from being insufficient.

The purpose of this study to minimize the position of any electric vehicle to any charging station at any time. In this way, it is ensured that the locations of electric vehicle charging stations are optimized. For this purpose, an MIT model was established and a solution was reached with the GAMS program.

1.3. Assumptions

Due to the increasing world population and spreading diseases (especially the COVID-19 virus), it has led people to use their personal vehicles instead of public transportation. This causes the number of vehicles to increase even more. As the technology develops over the years, the developments in the car industry will accelerate. Today, the production of vehicles that work with fossil fuels continues. However, as it is known, fossil fuels are non-renewable energy sources. Therefore, the fossil fuel resource in the world is decreasing day by day. With the developing technology, people have succeeded in producing cars that can run on electricity instead of fossil fuels. As fossil fuel resources decrease, their prices increase. Electric vehicles are becoming more and more popular day by day. It is claimed that the production of fossil fuel vehicles will end completely in the near future. This will greatly increase the number of electric vehicles. All these developments clearly show that the number of electric vehicle charging stations in the future will be insufficient and the waiting time at the stations will increase. Therefore, precautions should be taken without these problems. For all these reasons, location optimization should be done to find where the electric charging stations will be opened in the most accurate way.

2. APPLICATION

2.1. The Description of the Problem

The number of electric vehicle owners is increasing exponentially. Although it is not a big problem today, electric vehicle owners may have problems in charging their vehicles in the near future. Due to the long charging times, EV owners should be able to get to the nearest available charging station as soon as possible. Since the installation cost of charging stations is high, the number of stations should also be minimal. Therefore, the fact that the small number of charging stations is not positioned correctly creates problems.

2.2. Notations and the Data

Consider a finite time horizon T, divided into equal three periods (morning, noon, evening) indexed by t. During this time, a set of vehicles V, need to recharge at p charging stations chosen from a set of candidate locations J.

The set of vehicles V consists of three vehicles: car1, car2, car3. The set of candidate locations J is as follows: Başakşehir, Beşiktaş, Kadıköy, Çekmeköy.

The positions of the vehicles in each period are as follows:

In the Morning (Locations)		
car1	**car2**	**car3**
Bakırköy	Esenler	Maltepe

In the Noon (Locations)		
car1	**car2**	**car3**
Fatih	Üsküdar	Ümraniye

In the Evening (Locations)		
car1	**car2**	**car3**
Bakırköy	Ataşehir	Sancaktepe

Figure 2.2.1. *Locations of the vehicles*

The time after the position of the vehicles in each time zone their distances from the location they will go to are shown in the N matrix with $\mathbf{N_{vt}}$ elements. (Fig. 2).



Figure 2.2.2. *N Matrix*

The distance between the position of v in period t and the position of candidate station j is shown in D tensor with $\mathbf{d_{vtj}}$ elements.



Figure 2.2.3. *D Tensor*

If the vehicle v does not have enough charge to go to the desired location in period t, the distance between the candidate charging stations and the location it wants to go shown in E tensor with $\mathbf{E_{vtj}}$ elements.



Figure 2.2.4. *E Tensor*

Each candidate location j has a limited capacity and is denoted by $\mathbf{Q_j}$. In other words, a certain number of vehicles can be charged simultaneously at each candidate charging station. The number of vehicles that can be charged at the same time at the Başakşehir location is 8 ($Q_{Başahşehir}$ = 0). The capacities of Beşiktaş and Kadıköy candidate stations are 15 ($Q_{Beşiktaş}$ = $Q_{Kadıköy}$ = 15). At the last candidate station, Çekmeköy, the number of vehicles that can be charged at the same time is 10 ($Q_{Çekmeköy}$ = 10).

The batteries of the vehicles have a capacity (indicates the range in km) and this capacity is shown in $\mathbf{K_v}$. The capacity of all three vehicles is 100 kilometers ($K_{car1} = K_{car2} = K_{car3} = 100$).

The minimum number of charging stations that need to be opened for the system to work properly is expressed with **p**. Whether or not the charging station is opened in a location is checked in binary (0,1) and shown with $\mathbf{y_j}$. Likewise, whether a vehicle is in any location at any time or not is specified in the model in binary (0,1) with the variable $\mathbf{x_{vtj}}$. In addition, the amount of charge (range in kilometers) that the vehicle v has in the period t is shown with $\mathbf{R_{vt}}$, while the amount of charge (range in kilometers) it receives by visiting any station is also indicated by $\mathbf{a_{vt}}$. The initial R_{vt} values of car1, car2 and car3 are respectively as follows: 20, 50, 60

2.3. The Optimization Model

In this part, a new and original model is created by replacing the missing parts of the model proposed by Andrews [10] and the models proposed by Arayıcı and Poyrazoğlu [11], who contributed to the development of this model.

The MIP model, which ensures that the stations to be opened by using all these data are at the closest distance to the vehicles, is as follows.

(Z)	**Min.**	$\sum_{vtj} d_{vtj}\, x_{vtj}$	
(1)	**s.t.**	$\sum_j x_{vtj} \leq 1$,	$\forall v \in V, t \in T$
(2)		$\sum_{tj} x_{vtj} \geq 1$,	$\forall v \in V$
(3)		$x_{vtj} \leq y_j$,	$\forall v \in V, t \in T, j \in J$
(4)		$\sum_v x_{vtj} \leq Q_j\, y_j$,	$\forall j \in J, t \in T$
(5)		$\sum_{vtj} x_{vtj} \leq p$,	
(6)		$R_{vt} + a_{vt} - d_{vtj}\, x_{vtj} \leq K_V$,	$\forall v \in V, t \in T, j \in J$
(7)		$a_{vt} \geq 0$,	$\forall v \in V, t \in T$
(8)		$R_{vt} \geq d_{vtj}\, x_{vtj}$,	$\forall v \in V, t \in T, j \in J$
(9)		$a_{vt} \leq K_v \sum_j x_{vtj}$,	$\forall v \in V, t \in T$
(10)		$R_{vt} + a_{vt} - \left[\sum_j (E_{vtj} + d_{vtj}) x_{vtj}\right] - N_{vt}\left(1 - \sum_j x_{vtj}\right) \geq 0$,	$\forall v \in V, t \in T$
(11)		$R_{v(t+1)} = R_{vt} + a_{vt} - \left[\sum_j (E_{vtj} + d_{vtj}) x_{vtj}\right] - N_{vt}\,(1 - \sum_j x_{vtj})$,	$\forall v \in V, j \in J, \forall t\!: \{1, \ldots T-1\}$
(12)		$R_{(v=1)(t=1)} = R\_initial\_Car1$,	$\forall v\!: 1,\ \forall t\!: 1$
(13)		$R_{(v=2)(t=1)} = R\_initial\_Car2$,	$\forall v$:2, $\forall t$:1
(14)		$R_{(v=3)(t=1)} = R\_initial\_Car3$,	$\forall v$:3, $\forall t\!: 1$
(15)		$x_{vtj} \in \{0,1\}$,	$\forall v \in V, t \in T, j \in J$
(16)		$y_j \in \{0,1\}$,	$\forall j \in J$

The objective (Z) is to minimize the total distance traveled by all vehicles to access the selected charging stations. Constraint (1) ensures that each vehicle is charged by selecting one charging option. Constraint (2) ensures that each vehicle is charged at least once a day. Constraint (3) says vehicle v can charge at location j only if a charging station is opened there. In constraint (4), we state that its capacity must be suitable for us to be able to charge from a station opened at position j. The variable p in constraint (5) represents the minimum number of charging stations and ensures that the number of stations to be opened is minimal. Constraint (6) states that no matter how much the vehicle is charged, the amount of charge cannot exceed the maximum capacity of the vehicle. Constraint (7) ensures that the vehicle cannot discharge, that is, it can only charge. Constraint (8) states that in order for the vehicle to go to the location, the vehicle's range must not be less than its distance from the location. Constraint (9) ensures that the charge value a_{vt} is 0 if we are not going to visit location j. Constraint (10) checks whether the current range of our vehicle is enough to the target location, if not, it indicates that we should go to the charging station. Constraint (11) says our current range to decrease by the distance we travel, and if we get a charge from a station it adds it to our current range. Constraints (12), (13) and (14) determine the initial available range amounts of car1, car2 and car3, respectively. Constraints (15) and (16) ensure that our decision variables x_{vtj} and y_j only take values of 0 or 1.

2.4. GAMS Codes

Sets, scalar and parameters are defined:

```
sets
v vehicles /car1, car2, car3/
t time horizon /morning,noon,evening/
j candidate locations /basaksehir, besiktas, kadikoy, cekmekoy/
;

scalar
R_initial_Car1 amount of initial charge (km) of vehicle car1 /20/
R_initial_Car2 amount of initial charge (km) of vehicle car2 /50/
R_initial_Car3 amount of initial charge (km) of vehicle car3 /60/
;


parameters
q(j) capacity of the candidate locations
/basaksehir 8
 besiktas 15
 kadikoy 15
 cekmekoy 10/

k(v) battery capacity (km) of the vehicles
/car1 100
 car2 100
 car3 100/;
```

Figure 2.4.1. *GAMS code: Sets, scalar, parameters*

We define the N matrix with N_{vt}, the D tensor with d_{vtj} and E tensor with E_{vtj} as tables:

```

table n(v,t) distance to target location
      morning    noon    evening
car1  16         10      5
car2  35         11      38
car3  21         31      17
;

table d(v,t,j) the distance between the location of v in period t and location j
                basaksehir  besiktas  kadikoy  cekmekoy
car1.morning    15          21        25       56
car1.noon       15          15        18       49
car1.evening    15          21        25       56
car2.morning    10          25        38       52
car2.noon       34          12        8        17
car2.evening    43          20        5        33
car3.morning    53          26        12       29
car3.noon       40          14        11       29
car3.evening    60          38        31       17 ;

table e(v,t,j) the distance between location j and destination
                basaksehir  besiktas  kadikoy  cekmekoy
car1.morning    16          15        17       52
car1.noon       16          21        25       58
car1.evening    16          21        25       58
car2.morning    44          12        13       36
car2.noon       44          25        7        36
car2.evening    12          21        38       54
car3.morning    38          19        11       30
car3.noon       59          40        35       19
car3.evening    52          36        20       29 ;

```

Figure 2.4.2. *GAMS code: tables*

Variables are defined:

```
variable
Z objective function ;

positive variable
r(v,t) amount of current charge (km) of vehicle v in period t
a(v,t) amount of charge (km) that vehicle v receives in period t
p The minimum number of the electic charge stations
;

binary variable
x(v,t,j) whether v vehicle is charging in period
y(j) whether to open a charging station at candidate location j
;

```

Figure 2.4.3. *GAMS code: variables*

We define our objective function and constraints under the name “equation”:

```
equation
objective To minimize the distance of any vehicle from any charging station at any time.
constraint1 A vehicle cannot be charged at more than one station at the same time.
constraint2 Every vehicle must be charged at least once a day.
constraint3 In order for charging station must first be opened at that location.
constraint4 If the capacity of the station is full vehicles cannot be charged there.
constraint5 The number of stations opened should be the minimum amount.
constraint6 Charge received from a station cannot exceed the charge capacity of the vehicle
constraint7 Vehicles cannot be discharged
constraint8 Vehicles should be able to go to the station if needed.
constraint9 The charge value is 0 if we are not going to visit location j
constraint10 If the range is not enough it indicates that we have to go to the charging station
constraint11 Current range to decrease by the distance we travel
constraint12 The initial available range amounts of car1
constraint13 The initial available range amounts of car2
constraint14 The initial available range amounts of car3
;

```

Figure 2.4.4. *GAMS code: variables*

This is the main part where we code the model. The objective function and constraints are formulated with equations and inequalities:

```

objective.. Z =e= sum((v,t,j),d(v,t,j)*x(v,t,j));

constraint1(v,t).. sum(j,x(v,t,j)) =l= 1;
constraint2(v).. sum((t,j),x(v,t,j)) =g= 1;
constraint3(v,t,j).. x(v,t,j) =l= y(j);
constraint4(t,j).. sum(v,x(v,t,j)) =l= q(j)*y(j);
constraint5.. sum((v,t,j),x(v,t,j)) =l= p;
constraint6(v,t,j).. r(v,t)+a(v,t)-(d(v,t,j)*x(v,t,j)) =l= k(v);
constraint7(v,t).. a(v,t) =g= 0;
constraint8(v,t,j).. r(v,t) =g= d(v,t,j)*x(v,t,j);
constraint9(v,t).. a(v,t) =l= k(v)*sum(j,x(v,t,j));
constraint10(v,t).. r(v,t)+a(v,t)-sum(j,((e(v,t,j)+d(v,t,j))*x(v,t,j)))-(n(v,t)*(1-sum(j,x(v,t,j)))) =g= 0;
constraint11(v,t+1).. r(v,t+1) =e= r(v,t)+a(v,t)-sum(j,((e(v,t,j)+d(v,t,j))*x(v,t,j)))-(n(v,t)*(1-sum(j,x(v,t,j)))) ;
constraint12.. r('car1','morning') =e= R_initial_Car1;
constraint13.. r('car2','morning') =e= R_initial_Car2;
constraint14.. r('car3','morning') =e= R_initial_Car3;


model LOCOPT /all/;
solve LOCOPT using MIP minimizing Z;
display x.l,r.l,a.l,p.l,y.l;

```

Figure 2.4.5. *GAMS code: Objective and constraints*

When we run the model we coded in GAMS, we see that the model works successfully and we have minimized our objective function. The objective function Z is found to be 34, which is the minimum number for the feasible region but the values that interest us are the x_{vtj} values. If the vehicle visits v candidate station j, the d_{vtj} value is 1. So, vehicle v receives charge from candidate station j, we decide that station j should be opened. In the next part of the report, you can see the numerical results of the model and the candidate stations that were decided to be opened.

2.5. Numerical Results According to GAMS Report

According to the GAMS report, the d_{vtj} values are found as follows:

Başakşehir			
	Morning	Noon	Evening
Car1	1	0	0
Car2	0	0	0
Car3	0	0	0

Beşiktaş			
	Morning	Noon	Evening
Car1	0	0	0
Car2	0	0	0
Car3	0	0	0

Kadıköy			
	Morning	Noon	Evening
Car1	0	0	0
Car2	0	1	0
Car3	0	1	0

Çekmeköy			
	Morning	Noon	Evening
Car1	0	0	0
Car2	0	0	0
Car3	0	0	0

Figure 2.5.1. *Charging status at candidate locations in each period*

Car1 received a charge from the candidate station Başakşehir in the morning period. On the other hand, car2 and car3 received a charge from the candidate station Kadıköy in the noon period. According to these results, it should be decided to open the candidate stations Kadıköy and Başakşehir.

Finally, according to the final report, the current charge amounts of the vehicles in each period, the kilometers traveled by the vehicles and the amount of charge they received from the station were found as follows:



Figure 2.5.2. *Model results*

3. DISCUSSION AND FUTURE RESEARCH

As mentioned before, the model was developed in the GAMS program and successfully achieved using MIP.

The model was created by dividing a day equally into three time zones (morning, noon and evening), choosing four possible charging stations and using three electric vehicles. In general, the model determines the most likely charging points at the shortest distance according to the vehicles moving in different scenarios on a certain route.

As a result of the entered data and constraints, the optimization model has successfully shown the results showing the state of charge of the vehicles and how far they have traveled in each time period. It also enabled the determination of how much was charged in the charging case made in any time period. The range of the vehicles in each time period, the distance of the road to be traveled while passing to the next time period and if the vehicle needs to be charged in order to travel this road, the amount of charge made has been determined.

The developed model achieves the best results regardless of the data size. In case the data grows instead of the test data used in this study, the result can be achieved successfully without the need for any editing on the model.

This study is a small step of a large-scale project. In order to solve our problem comprehensively; The density of the existing parking lots and the waiting times of the vehicles should be analyzed. The distance of the car parks to the closest transformer center and the capacity of the transformer center should be determined. The number and power of the charging unit should be chosen considering the traffic and usage density, as well as the power of the transformer center and the amount of loading. The weekly, daily and hourly load demand curves of the point where the charging units will be connected should be obtained and analyzed. The distance of the parking lot to the closest transformer station and the capacity of the transformer center should be determined. The number and power of the charging unit should be chosen considering the traffic and usage density, as well as the power of the transformer center and the amount of loading. The weekly, daily and hourly load demand curves of the point where the charging units will be connected should be obtained and analyzed. While the planning and capacity of the new car parks to be built are examined in terms of traffic and transportation parameters, a parking lot planning including a charging unit should be made, taking into account the density and increasing number of electric vehicles in traffic.

REFERENCES

[1] Ministry of Public Security of the People's Republic of China. Available online: http://www.mps.gov.cn/n2254098/n4904352/c6354939/content.html (accessed on 13 May 2019)

[2] K. Schneider, C. Gerkensmeyer, M. Kintner-Meyer, and R. Fletcher. Impact assessment of plug-in hybrid vehicles on pacific northwest distribution systems. In Proceedings of IEEE Power and Energy Society General Meeting (PES), pages 1–6, 2008.

[3] Yağcıtekin B.; Elektrikli Araç Şarj Altyapısı Tasarımı ve Akıllı Şarj Sisteminin Geliştirilmesi, 2014.

[4] Vazifeh, M.M.; Zhang, H.; Santi, P.; Ratti, C. Optimizing the deployment of electric vehicle charging stations using pervasive mobility data. Transp. Res. Part A Policy Pr. 2019, 121, 75–91.

[5] Rahman, I.; Vasant, P.M.; Singh, B.S.M.; Abdullah-Al-Wadud, M. Swarm Intelligence-Based Smart Energy Allocation Strategy for Charging Stations of Plug-In Hybrid Electric Vehicles. Math. Probl. Eng. 2015, 2015, 1–10.

[6] Vasant, P.M.; Rahman, I.; Singh, B.S.M.; Abdullah-Al-Wadud, M. Optimal power allocation scheme for plug-in hybrid electric vehicles using swarm intelligence techniques. Cogent Eng. 2016, 3.

[7] Fredriksson H.; Dahl M.; Holmgren J. Optimal placement of Charging Stations for Electric Vehicles in large-scale Transportation Networks, 2019.

[8] Xiong Y.; Gan J.; An B.; Miao C.; Bazzan A. Optimal Electric Vehicle Charging Station Placement,2015.

[9] Bozkurt E. Elektrikli Araçlarda Şarj Durumuna Bağlı Optimum Rota Seçimi, 2011.

[10] M. Andrews, M. K. Dogru, J. D. Hobby, G. H. Tucci, and Y. Jin, "Modeling and Optimization for Electric Vehicle Charging Infrastructure," Alcatel-Lucent Bell Labs, 2012.